1. Overview

This document provides a complete overall picture of bridging between Grafcet, Petri Net and PNML. It also provides an overview of the whole transformation sequence that enables to produce an XML Petri net representation (in the PNML format [1]) from a textual definition of a Grafcet and in the opposite way.

So this document describes how bridges between Grafcet, Petri Net and PNML have been built, using a model transformation language ATL. This construction is composed of five steps:

- Grafcet Models conforming to its metamodel are injected from a textual definition of the grafcet by means of a TCS (Textual Concrete Syntax) program (this part is out of the scope of the document).
- A transformation from Grafcet in their Petri Net equivalent and inverse: the Grafcet – Petri Net Bridge.
- A transformation from Petri Net generated with Grafcet in their PNML equivalent and inverse: the Petri Net - PNML Bridge.
- A transformation from PNML generated with Petri Net in their XML equivalent and inverse: the PNML - XML Bridge.
- As a final step, the XML model is extracted to the textual XML representation using an ATL query.

The next sections will explain the different steps to realize these bridges between Grafcet, Petri Net, PNML and XML. Section 2 presents all metamodels; Section 3 explains all bridges and their transformations.



Figure 1 - ATL project overview

2. Metamodels

2.1. Grafcet

2.1.1. Generalities about Grafcet

Grafcet is a mainly French-based representation support for discrete system. It is a mode of representation and analysis of an automatism, particularly adapted to sequential systems with evolution, i.e. decomposable in steps. The Grafcet's name came from «graph» because this model had a graphic basis, and AFCET (Association française de cybernétique économique et technique) from the scientific association which supported it. The Grafcet represents graphically the operation of an automatism by: steps with associated action, transitions between steps, and directed connections between the steps and the transitions.

2.1.2. A simplified metamodel of Grafcet



Figure 2 - Grafcet metamodel

Description of this metamodel:

- "Grafcet": the main or root element which represent a grafcet,
- It is composed of elements and connections which are abstract class,
- Elements are "Step" or "Transition",
- Connections are "StepToTransition" or "TransitionToStep",
- Steps and transitions can have many incoming or outgoing connections.

2.2. Petri Net

2.2.1. Generalities about Petri Net

Petri nets are also known as a place/transition net or P/T net. Defined in 1962 by Carl Adam Petri, they extend state machines with a notion of concurrency. It is a graphical and mathematical representation of discrete distributed systems. Petri nets consist of places, transitions and directed arcs that connect them, so arcs run between places and transitions, not between places and places or transitions and transitions. There are two sorts of arcs connecting place to transition or transition to place.

2.2.2. A simplified metamodel of Petri Net



Figure 3 - Petri Net metamodel

Description of the basic metamodel:

- "PetriNet": the main or root element which represent a Petri net,
- It is composed of elements and arcs which are abstract class,
- Elements are "Place" or "Transition",
- Arcs are "PlaceToTransition" or "TransitionToPlace",
- Places and transitions can have many incoming or outgoing arcs.

2.3. PNML

2.3.1. Generalities about PNML

The Petri Net Markup Language (PNML) is a proposal of an XML-based interchange format for Petri nets (see [1]). Originally, it was intended to serve as a file format for the Java version of the Petri Net Kernel. PNML is a concept for defining the overall structure of a Petri net file.

2.3.2. A simplified metamodel of PNML



Figure 4 - PNML metamodel

Description of the simplified metamodel:

- "PNMLDocument": the main or root element which contains Petri nets,
- "NetElement" represents the Petri net; it is composed of "NetContent" which are "Arc", "Place" and "Transition".
- Arcs reference a source and a target ("Place" or "Transition") but the two kinds of arcs are not differentiated in this model (PlaceToTransition and TransitionToPlace).
- Net elements and net contents can have a name which is a labelled element composed of labels.

2.4. XML

The XML metamodel describes the different model elements that compose a XML model, as well as the way they can be linked to each other. The considered metamodel is presented in Figure 7. It is moreover provided in KM3 format [2] in Appendix V.



Figure 5 - XML metamodel

Description of the basic metamodel:

A XML model has a single Root element. It also contains Elements, Texts, Attributes entities. The Attribute, Text and Element elements all directly inherit from the abstract Node element, whereas Root inherits from the Element entity. The following attributes are defined for the abstract Node entity: “startLine”, “startColumn”, “endLine”, “endColumn”, “name” and “value”. In the scope of this example, we only make use of the two last attributes, “name” and “value”. In case of an Attribute entity, “name” encodes the name of the attribute, whereas “value” contains the value associated with the Attribute. In case of a Text entity, “value” contains the textual content of the Text. Finally, considering an Element entity, “name” encodes the name of the modelled XML tag.

An Element can contain several Nodes, which can be either of type Attribute, Text or Element. Inversely, a Node can be contained by zero or one Element. In fact, each Node is contained by an Element except the Root element which has no parent.

3. Bridges

3.1. The Grafcet – Petri Net Bridge

3.1.1. Grafcet to Petri Net Transformation

In order to realize the bridge, and as there is no markup language for Grafcet, we need a textual input file. So, the Grafcet model is imported to a textual representation by means of a TCS (Textual Concrete Syntax) program. This part is not documented in this document.

3.1.1.1. Description of the Transformation

This transformation takes a Grafcet model conforming to our Grafcet metamodel and maps all Grafcet's features to Petri Net. In fact the two metamodels of Grafcet and Petri Net are very close. So this transformation is quite easy. The ATL code for this transformation consists of 5 rules and no helpers.

Rules:

- The **PetriNet** rule generates a PetriNet element from the input Grafcet element. The name of the generated PetriNet element is copied from the one of the input Grafcet. Its set of Elements corresponds to Elements generated by Place and Transition rules. And its set of Arcs corresponds to Connections generated by PlaceToTransition and TransitionToPlace rules.
- The **Place** rule generates a Place element from the input Step element. The name of the generated Place element is copied from the one of the input Step. Its set of incomingArcs corresponds to incomingConnections generated by TransitionToPlace rule. And its set of outgoingArc corresponds to outgoingConnections generated by PlaceToTransition rule.
- The **Transition** rule generates a Transition element from the input Transition element. The name of the generated Transition element is copied from the one of the input Transition. Its set of incomingArcs corresponds to incomingConnections generated by PlaceToTransition rule. And its set of outgoingArc corresponds to outgoingConnections generated by TransitionToPlace rule.
- The **PlaceToTransition** rule generates a PlaceToTransition element from the input StepToTransition element. The name of the generated PlaceToTransition element is copied from the one of the input StepToTransition. Its *from* and *to* references are also copied from the ones of the input StepToTransition.
- The **TransitionToPlace** rule generates a TransitionToPlace element from the input TransitionToStep element. The name of the generated TransitionToPlace element is copied from the one of the input TransitionToStep. Its *from* and *to* references are also copied from the ones of the input TransitionToStep.

3.1.1.2. ATL Code

```
1   module Grafcet2PetriNet;
2   create OUT : PetriNet from IN : Grafcet;
3
4   -- The PetriNet rule generates a PetriNet element from the input Grafcet
5   element.
6   -- Name of the generated PetriNet element is copied from the one of the
7   input Grafcet.
8   -- Its set of Elements corresponds to Elements generated by Place and
9   Transition rules.
10  -- And its set of Arcs corresponds to Connections generated by
11  PlaceToTransition and TransitionToPlace rules.
12  rule PetriNet {
13    from
14      g : Grafcet!Grafcet
15    to
16      p : PetriNet!PetriNet
17      (
18        location <- g.location,
19        name <- g.name,
20        elements <- g.elements,
21        arcs <- g.connections
22      )
23  }
24
25  -- The Place rule generates a Place element from the input Step element.
26  -- Name of the generated Place element is copied from the one of the input
27  Step.
28  -- Its set of incomingArcs corresponds to incomingConnections generated by
29  TransitionToPlace rule.
30  -- And its set of outgoingArc corresponds to outgoingConnections generated
31  by PlaceToTransition rule.
32  rule Place {
33    from
34      g : Grafcet!Step
35    to
36      p : PetriNet!Place
37      (
38        location <- g.location,
39        name <- g.name,
40        net <- g.grafcet,
41        incomingArc <- g.incomingConnections,
42        outgoingArc <- g.outgoingConnections
43      )
44  }
45
46  -- The Transition rule generates a Transition element from the input
47  Transition element.
48  -- Name of the generated Transition element is copied from the one of the
49  input Transition.
50  -- Its set of incomingArcs corresponds to incomingConnections generated by
51  PlaceToTransition rule.
```

```
52  -- And its set of outgoingArc corresponds to outgoingConnections generated
53  by TransitionToPlace rule.
54  rule Transition {
55    from
56      g : Grafcet!Transition
57    to
58      p : PetriNet!Transition
59      (
60        location <- g.location,
61        name <- g.name,
62        net <- g.grafcet,
63        incomingArc <- g.incomingConnections,
64        outgoingArc <- g.outgoingConnections
65      )
66  }
67
68  -- The PlaceToTransition rule generates a PlaceToTransition element from
69  the input StepToTransition element.
70  -- Name of the generated PlaceToTransition element is copied from the one
71  of the input StepToTransition.
72  -- Its from and to references are also copied from the ones of the input
73  StepToTransition.
74  rule PlaceToTransition {
75    from
76      g : Grafcet!StepToTransition
77    to
78      p : PetriNet!PlaceToTransition
79      (
80        location <- g.location,
81        name <- g.name,
82        net <- g.grafcet,
83        "from" <- g."from",
84        "to" <- g."to"
85      )
86  }
87  -- The TransitionToPlace rule generates a TransitionToPlace element from
88  the input TransitionToStep element.
89  -- Name of the generated TransitionToPlace element is copied from the one
90  of the input TransitionToStep.
91  -- Its from and to references are also copied from the ones of the input
92  TransitionToStep.
93  rule TransitionToPlace {
94    from
95      g : Grafcet!TransitionToStep
96    to
97      p : PetriNet!TransitionToPlace
98      (
99        location <- g.location,
100       name <- g.name,
101       net <- g.grafcet,
102       "from" <- g."from",
103       "to" <- g."to"
104     )
105 }
```

3.1.1.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 6 and Figure 7, there is one input metamodel (Grafcet) and one output (Petri Net). In Path Editor, place in "Grafcet" the path of the Grafcet metamodel; do the same for "PetriNet". In "IN" place the path of an Ecore file (a model conforming to our Grafcet metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the Petri Net metamodel).



Figure 6 - Grafcet to Petri Net configuration - part one



Figure 7 - Grafcet to Petri Net configuration - part two

3.1.2. Petri Net to Grafcet Transformation

3.1.2.1. Description of the Transformation

As two metamodels of Grafcet and Petri Net are very close, this transformation is very similar to the previous one. The ATL code for the Petri Net to Grafcet transformation also consists of 5 rules and no helpers. All the rules are identical, only the input elements became output elements and in the reverse way.

Rules:

- The **Grafcet** rule generates a Grafcet element from the input Petri Net element. The name of the generated Grafcet element is copied from the one of the input Petri Net. Its set of Elements corresponds to Elements generated by Step and Transition rules. And its set of Connections corresponds to Arcs generated by StepToTransition and TransitionToStep rules.
- The **Step** rule generates a Step element from the input Place element. The name of the generated Step element is copied from the one of the input Place. Its set of incomingConnections corresponds to incomingArcs generated by TransitionToStep rule. And its set of outgoingConnections corresponds to outgoingArc generated by StepToTransition rule.
- The **Transition** rule generates a Transition element from the input Transition element. The name of the generated Transition element is copied from the one of the input Transition. Its set of incomingConnections corresponds to incomingArcs generated by StepToTransition rule. And its set of outgoingConnections corresponds to outgoingArc generated by TransitionToStep rule.
- The **StepToTransition** rule generates a StepToTransition element from the input PlaceToTransition element. The name of the generated StepToTransition element is copied from the one of the input PlaceToTransition. Its *from* and *to* references are also copied from the ones of the input PlaceToTransition.
- The **TransitionToStep** rule generates a TransitionToStep element from the input TransitionToPlace element. The name of the generated TransitionToStep element is copied from the one of the input TransitionToPlace. Its *from* and *to* references are also copied from the ones of the input TransitionToPlace.

3.1.2.2. ATL Code

```
1  module PetriNet2Grafcet;
2  create OUT : Grafcet from IN : PetriNet;
3
4  -- The Grafcet rule generates a Grafcet element from the input Petri Net
5  element.
6  -- Name of the generated Grafcet element is copied from the one of the
7  input Petri Net.
8  -- Its set of Elements corresponds to Elements generated by Step and
9  Transition rules.
10 -- And its set of Connections corresponds to Arcs generated by
11 StepToTransition and TransitionToStep rules.
12 rule Grafcet {
13   from
14     p : PetriNet!PetriNet
15
16   to g : Grafcet!Grafcet
17     (
18       location <- p.location,
19       name <- p.name,
20       elements <- p.elements,
21       connections <- p.arcs
22     )
23 }
```

```
24
25  -- The Step rule generates a Step element from the input Place element.
26  -- Name of the generated Step element is copied from the one of the input
27  Place.
28  -- Its set of incomingConnections corresponds to incomingArcs generated by
29  TransitionToStep rule.
30  -- And its set of outgoingConnections corresponds to outgoingArc generated
31  by StepToTransition rule.
32  rule Step {
33    from
34      p : PetriNet!Place
35    to
36      g : Grafcet!Step
37      (
38        location <- p.location,
39        name <- p.name,
40        grafcet <- p.net,
41        isInitial <- false,
42        isActive <- false,
43        incomingConnections <- p.incomingArc,
44        outgoingConnections <- p.outgoingArc
45      )
46  }
47
48  -- The Transition rule generates a Transition element from the input
49  Transition element.
50  -- Name of the generated Transition element is copied from the one of the
51  input Transition.
52  -- Its set of incomingConnections corresponds to incomingArcs generated by
53  StepToTransition rule.
54  -- And its set of outgoingConnections corresponds to outgoingArc generated
55  by TransitionToStep rule.
56  rule Transition {
57    from
58      p : PetriNet!Transition
59
60    to
61      g : Grafcet!Transition
62      (
63        location <- p.location,
64        name <- p.name,
65        grafcet <- p.net,
66        incomingConnections <- p.incomingArc,
67        outgoingConnections <- p.outgoingArc
68      )
69  }
70
71  -- The StepToTransition rule generates a StepToTransition element from the
72  input PlaceToTransition element.
73  -- Name of the generated StepToTransition element is copied from the one of
74  the input PlaceToTransition.
75  -- Its from and to references are also copied from the ones of the input
76  PlaceToTransition.
77  rule StepToTransition {
```

```
78      from
79        p : PetriNet!PlaceToTransition
80      to
81        g : Grafcet!StepToTransition
82        (
83           location <- p.location,
84           name <- p.name,
85           grafcet <- p.net,
86           "from" <- p."from",
87           "to" <- p."to"
88        )
89    }
90
91    -- The TransitionToStep rule generates a TransitionToStep element from the
92    input TransitionToPlace element.
93    -- Name of the generated TransitionToStep element is copied from the one of
94    the input TransitionToPlace.
95    -- Its from and to references are also copied from the ones of the input
96    TransitionToPlace.
97    rule TransitionToStep {
98      from
99        p : PetriNet!TransitionToPlace
100     to
101       g : Grafcet!TransitionToStep
102       (
103          location <- p.location,
104          name <- p.name,
105          grafcet <- p.net,
106          "from" <- p."from",
107          "to" <- p."to"
108       )
109   }
```

3.1.2.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 8 and Figure 9, there is one input metamodel (Petri Net) and one output (Grafcet). In Path Editor, place in "PetriNet" the path of the Petri net metamodel; do the same for "Grafcet". In "IN" place the path of an Ecore file (a model conforming to our Petri net metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the Grafcet metamodel).



Figure 8 - Petri Net to Grafcet configuration - part one



Figure 9 - Petri Net to Grafcet configuration - part two

3.2. The Petri Net - PNML Bridge

3.2.1. Petri Net to PNML Transformation

3.2.1.1. Description of the Transformation

The ATL code for the Grafcet to Petri Net transformation consists of 4 rules and no helpers. In fact the two metamodels of Grafcet and Petri Net are quite close. So this transformation is quite easy.

Rules:

- The **PNMLDocument** rule generates a PNMLDocument and the NetElement which corresponds to the input PetriNet element. The name of the generated NetElement is copied from the one of the input PetriNet, by creating a PNML Name composed of a PNML Label which value is initialized by the PetriNet name. Its set of Contents corresponds to the union of the PetriNet Elements and Arcs.
- The **Place** rule generates a Place corresponds to the input PetriNet Place element. The name of the generated Place is copied from the one of the input Place, by creating a PNML Name composed of a PNML Label which value is initialized by the PetriNet Place name.
- The **Transition** rule generates a Transition corresponds to the input PetriNet Transition element. The name of the generated Transition is copied from the one of the input Transition, by creating a PNML Name composed of a PNML Label which value is initialized by the PetriNet Transition name.

- The **Arc** rule generates a Arc corresponds to the input PetriNet Arc element (TransitionToPlace and PlaceToTransition). The name of the generated Arc is copied from the one of the input Arc, by creating a PNML Name composed of a PNML Label which value is initialized by the PetriNet Arc name. Its *source* and *target* references are also copied from the input Arc and correspond respectively to *from* and *to* references.

3.2.1.2. ATL Code

```
1   module PetriNet2PNML;
2   create OUT : PNML from IN : PetriNet;
3
4   -- The PNMLDocument rule generates a PNMLDocument  and the NetElement which
5   corresponds to the input PetriNet element.
6   -- Name of the generated NetElement is copied from the one of the input
7   PetriNet, by creating a PNML Name composed of a PNML Label which value is
8   initialized by the PetriNet name.
9   -- Its set of Contents corresponds to the union of the PetriNet Elements
10  and Arcs.
11  rule PNMLDocument {
12    from
13      e : PetriNet!PetriNet
14    to
15      n : PNML!PNMLDocument
16      (
17        location <- e.location,
18        xmlns <- uri,
19        nets <- net
20      ),
21      uri : PNML!URI
22      (
23        value <- 'http://www.informatik.hu-berlin.de/top/pnml/ptNetb'
24      ),
25      net : PNML!NetElement
26      (
27        name <- name,
28        location <- e.location,
29        id <- e.location,
30        type <- type_uri,
31        contents <- e.elements.union(e.arcs)
32      ),
33      name : PNML!Name
34      (
35        labels <- label
36      ),
37      label : PNML!Label
38      (
39        text <- e.name
40      ),
41      type_uri : PNML!URI
42      (
43        value <- 'http://www.informatik.hu-berlin.de/top/pntd/ptNetb'
44      )
45  }
46
```

```
47  -- The Place rule generates a Place corresponds to the input PetriNet Place
48  element.
49  -- Name of the generated Place is copied from the one of the input Place,
50  by creating a PNML Name composed of a PNML Label which value is initialized
51  by the PetriNet Place name.
52  rule Place {
53    from
54      e : PetriNet!Place
55    to
56      n : PNML!Place
57      (
58        name <- name,
59        id <- e.name,
60        location <- e.location
61      ),
62      name : PNML!Name
63      (
64        labels <- label
65      ),
66      label : PNML!Label
67      (
68        text <- e.name
69      )
70  }
71
72  -- The Transition rule generates a Transition corresponds to the input
73  PetriNet Transition element.
74  -- Name of the generated Transition is copied from the one of the input
75  Transition, by creating a PNML Name composed of a PNML Label which value is
76  initialized by the PetriNet Transition name.
77  rule Transition {
78    from
79      e : PetriNet!Transition
80    to
81      n : PNML!Transition
82      (
83        name <- name,
84        id <- e.name,
85        location <- e.location
86      ),
87      name : PNML!Name
88      (
89        labels <- label
90      ),
91      label : PNML!Label
92      (
93        text <- e.name
94      )
95  }
96
97  -- The Arc rule generates a Arc corresponds to the input PetriNet Arc
98  element (TransitionToPlace and PlaceToTransition).
```

```
99  -- Name of the generated Arc is copied from the one of the input Arc, by
100 creating a PNML Name composed of a PNML Label which value is initialized by
101 the PetriNet Arc name.
102 -- Its source and target references are also copied from the input Arc and
103 correspond respectively to the from and to references.
104 rule Arc {
105   from
106     e : PetriNet!Arc
107   to
108     n : PNML!Arc
109     (
110       name <- name,
111       location <- e.location,
112       id <- e.name,
113       source <- e."from",
114       target <- e."to"
115     ),
116     name : PNML!Name
117     (
118       labels <- label
119     ),
120     label : PNML!Label
121     (
122       text <- e.name
123     )
124 }
```

3.2.1.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 10 and Figure 11, there is one input metamodel (Petri Net) and one output (PNML). In Path Editor, place in "PetriNet" the path of the Petri net metamodel; do the same for "PNML". In "IN" place the path of an Ecore file (a model conforming to our Petri net metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the PNML metamodel).



Figure 10 - Petri Net to PNML configuration - part one



Figure 11 - Petri Net to PNML configuration - part two

3.2.2. PNML to Petri Net Transformation

3.2.2.1. Description of the Transformation

The ATL code for the Grafcet to Petri Net transformation consists of 5 rules and no helpers. In fact the two metamodels of Grafcet and Petri Net are quite close. So this transformation is quite easy.

Rules:

- The **PetriNet** rule generates a PetriNet which corresponds to the input NetElement included in the PNMLDocument. The name of the generated PetriNet is copied from the one of the input NetElement, by recovering the value of the PNML Label included in the PNML Name of the NetElement. Its set of Elements is the corresponding set named “elementsSet” calculated in the using clause. And its set of Arcs is the corresponding set named “arcsSet” calculated in the using clause.

- The **Place** rule generates a Place which corresponds to the input Place. The name of the generated Place is copied from the one of the input Place, by recovering the value of the PNML Label included in the PNML Name of the PNML Place.

- The **Transition** rule generates a Transition which corresponds to the input Transition. The name of the generated Transition is copied from the one of the input Transition, by recovering the value of the PNML Label included in the PNML Name of the PNML Transition.

- The **PlaceToTransition** rule generates a PlaceToTransition which corresponds to the input Arc which has a Place for source and a Transition for Target. The name of the generated PlaceToTransition is copied from the one of the input Arc, by recovering the value of the

PNML Label included in the PNML Name of the PNML Arc. Its *from* and *to* references are also copied from the input Arc and correspond respectively to the *source* and *target* references.

- The **TransitionToPlace** rule generates a TransitionToPlace which corresponds to the input Arc which has a Transition for source and a Place for Target. The name of the generated TransitionToPlace is copied from the one of the input Arc, by recovering the value of the PNML Label included in the PNML Name of the PNML Arc. Its *from* and *to* references are also copied from the input Arc and correspond respectively to the *source* and *target* references.

3.2.2.2. ATL Code

```
1  module PNML2PetriNet;
2  create OUT : PetriNet from IN : PNML;
3
4  -- The PetriNet rule generates a PetriNet which corresponds to the input
5  NetElement included in the PNMLDocument.
6  -- Name of the generated PetriNet is copied from the one of the input
7  NetElement, by recovering the value of the PNML Label included in the PNML
8  Name of the NetElement.
9  -- Its set of Elements is the corresponding set named “elementsSet”
10 calculated in the using clause.
11 -- And its set of Arcs is the corresponding set named “arcsSet” calculated
12 in the using clause.
13 rule PetriNet {
14   from
15     n : PNML!PNMLDocument
16   using{
17        elementsSet : Set(PetriNet!Element) =
18          PNML!NetContentElement.allInstances();
19
20        arcsSet : Set(PetriNet!Arc) =
21          PNML!Arc.allInstances();
22     }
23   to
24     p : PetriNet!PetriNet
25     (
26        location <- n.location,
27        name <- n.nets.first().name.labels.first().text,
28        elements <- elementsSet,
29        arcs <- arcsSet
30     )
31 }
32
33 -- The Place rule generates a Place which corresponds to the input Place.
34 -- Name of the generated Place is copied from the one of the input Place ,
35 by recovering the value of the PNML Label included in the PNML Name of the
36 PNML Place.
37 rule Place {
38   from
39     n : PNML!Place
40   to
41     p : PetriNet!Place
42     (
43        location <- n.location,
```

```
44          name <- n.name.labels.first().text,
45          net <- n.net.document
46       )
47    }
48
49    -- The Transition rule generates a Transition which corresponds to the
50    input Transition .
51    -- Name of the generated Transition is copied from the one of the input
52    Transition , by recovering the value of the PNML Label included in the PNML
53    Name of the PNML Transition .
54    rule Transition {
55       from
56          n : PNML!Transition
57       to
58          p : PetriNet!Transition
59          (
60             location <- n.location,
61             name <- n.name.labels.first().text,
62             net <- n.net.document
63          )
64    }
65
66    -- The PlaceToTransition rule generates a PlaceToTransition which
67    corresponds to the input Arc which has a Place for source and a Transition
68    for Target.
69    -- Name of the generated PlaceToTransition is copied from the one of the
70    input Arc, by recovering the value of the PNML Label included in the PNML
71    Name of the PNML Arc.
72    -- Its from and to references are also copied from the input Arc and
73    correspond respectively to the source and target references.
74    rule PlaceToTransition {
75       from
76          n : PNML!Arc
77          (  -- arc source must be a place and arc target a transition
78             n.source.oclIsKindOf(PNML!Place) and
79    n.target.oclIsKindOf(PNML!Transition)
80          )
81       to
82          p : PetriNet!PlaceToTransition
83          (
84             location <- n.location,
85             name <- n.name.labels.first().text,
86             net <- n.net.document,
87             "from" <- n.source,
88             "to" <- n.target
89          )
90    }
91
92    -- The TransitionToPlace rule generates a TransitionToPlace which
93    corresponds to the input Arc which has a Transition for source and a Place
94    for Target.
95    -- Name of the generated TransitionToPlace is copied from the one of the
96    input Arc, by recovering the value of the PNML Label included in the PNML
97    Name of the PNML Arc.
```

```
98  -- Its from and to references are also copied from the input Arc and
99  correspond respectively to the source and target references.
100 rule TransitionToPlace {
101   from
102     n : PNML!Arc
103     (  -- arc source must be a transition and arc target a place
104       n.source.oclIsKindOf(PNML!Transition) and
105 n.target.oclIsKindOf(PNML!Place)
106     )
107   to
108     p : PetriNet!TransitionToPlace
109     (
110       location <- n.location,
111       name <- n.name.labels.first().text,
112       net <- n.net.document,
113       "from" <- n.source,
114       "to" <- n.target
115     )
116 }
```

3.2.2.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 12 and Figure 13, there is one input metamodel (PNML) and one output (PetriNet). In Path Editor, place in "PetriNet" the path of the Petri net metamodel; do the same for "PNML". In "IN" place the path of an Ecore file (a model conforming to our PNML metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the Petri net metamodel).



Figure 12 - PNML to Petri Net configuration - part one



Figure 13 - PNML to Petri Net configuration - part two

3.3. The PNML - XML Bridge

3.3.1. PNML to XML Transformation: Extractor

3.3.1.1. Description of the Transformation

The ATL code for the Grafcet to Petri Net transformation consists of 4 rules and 1 helper.

Helper:

- The **getRoot** helper is a constant helper. It seeks the root element of PNML model: the PNML document. This helper allows to link elements and their parents, thanks to a “resolveTemp” instruction.

Rules:

- The **Root** rule generates the XML Root element as well as a collection of attributes and elements and Text node from the input PNMLDocument element. The generated Root element is a “pnml” tag that has an “xmlns” Attribute and a “net” Element as children. The value of the “xmlns” attribute is copied from the PNMLDocument. The “net” Element has an “id” and a “type” Attribute, a “name” sub-Element. The “id” attribute and the “type” attribute are also copied from the input element. Finally, the “name” Element contains a “text” Element, which itself contains a Text node whose value corresponds to the name of the input PNMLDocument element.

- The **Place** rule generates three XML Elements, one XML Attribute and one XML Text for each PNML Place input element. The first generated Element, “place”, is a “place” tag which

accepts an “id” Attribute as well as a child “name” Element. The value of the “id” attribute corresponds to the one of the PNML Place. The generated “name” Element accepts a “text” Element as child. This last one has a child which is a Text node. Its value corresponds to the name of the input Place.

- The **Transition** rule generates three XML Elements, one XML Attribute and one XML Text for each PNML Transition input element. The first generated Element, “transition”, is a “transition” tag which accepts an “id” Attribute as well as a child “name” Element. The value of the “id” attribute corresponds to the one of the PNML Transition. The generated “name” Element accepts a “text” Element as child. This last one has a child which is a Text node. Its value corresponds to the name of the input Transition.

- The **Arc** rule generates three XML Elements, three XML Attributes and one XML Text for each PNML Arc input element. The generated Element is an “arc” tag that has three Attribute children: “id”, “source” and “target”, as well as a child “name” Element. The value of the “id” attribute corresponds to the one of the PNML Arc. Values of the “source” and “target” attributes respectively correspond to the id of the source and the id of the target of the input Arc. The generated “name” Element accepts a “text” Element as child. This last one has a child which is a Text node. Its value corresponds to the name of the input Transition.

3.3.1.2. ATL Code

```
1  module PNML2XML;
2  create OUT : XML from IN : PNML;
3
4  -- The getRoot helper, is a constant helper. It seeks the root element of
5  PNML model : the PNML document.
6  -- This helper allows to link elements and their parents, thanks to a
7  “resolveTemp” instruction and to the helper.
8  -- CONTEXT: n/a
9  -- RETURN: PNML!PNMLDocument
10 helper def: getRoot() : PNML!PNMLDocument =
11   PNML!PNMLDocument.allInstances()->asSequence()->first();
12
13
14 -- The Root rule generates the XML Root element as well as a collection of
15 attributes and elements and Text node from the input PNMLDocument element.
16 The generated Root element is a “pnml” tag that has an “xmlns” Attribute
17 and a “net” Element as children.
18 -- Value of the “xmlns” attribute is copied from the PNMLDocument. The
19 “net” Element has an “id” and a “type” Attribute, a “name” sub-Element.The
20 “id” attribute and the “type” attribute are also copied from the input
21 element.
22 -- Finally, the “name” Element contains a “text” Element, which itself
23 contains a Text node whose value corresponds to the name of the input
24 PNMLDocument element.
25 rule Root {
26   from
27     n : PNML!PNMLDocument
28   to
29     e : XML!Root
30     (
```

```
31          name <- 'pnml',
32          -- value = name of the net contained by this document
33          value <- n.nets.first().name.labels.first().text,
34          children <- Sequence {document_name, document_xmlns, document_net}
35        ),
36        document_name : XML!Element
37        (
38          name <- 'name',
39          parent <- n,
40          children <- document_text
41        ),
42        document_text : XML!Element
43        (
44          name <- 'text',
45          parent <- document_name,
46          children <- document_xml_text
47        ),
48        document_xml_text : XML!Text
49        (
50          value <- n.nets.first().name.labels.first().text,
51          parent <- document_text
52        ),
53        document_xmlns : XML!Attribute
54        (
55          name <- 'xmlns',
56          value <- n.xmlns.value,
57          parent <- n
58        ),
59        document_net : XML!Element
60        (
61          name <-'net',
62          value <- n.nets.first().name.labels.first().text,
63          parent <- n,
64          children <- Sequence {net_name, net_id, net_type}
65        ),
66        net_name : XML!Element
67        (
68          name <- 'name',
69          parent <- document_net,
70          children <- net_text
71        ),
72        net_text : XML!Element
73        (
74          name <- 'text',
75          parent <- net_name,
76          children <- net_xml_text
77        ),
78        net_xml_text : XML!Text
79        (
80          value <- n.nets.first().name.labels.first().text,
81          parent <- net_text
82        ),
83        net_id : XML!Attribute
84        (
```

```
85          name <- 'id',
86          value <- n.nets.first().id,
87          parent <- document_net
88       ),
89       net_type : XML!Attribute
90       (
91          name <-'type',
92          value <- n.nets.first().type.value,
93          parent <- document_net
94       )
95    }
96
97    -- The Place rule generates three XML Elements, one XML Attribute and one
98    XML Text for each PNML Place input element.
99    -- The first generated Element, “place”, is a “place” tag which accepts an
100   “id” Attribute as well as a child “name” Element. The value of the “id”
101   attribute corresponds to the one of the PNML Place.
102   -- The generated “name” Element accepts a “text” Element as child. This
103   last one has a child which is a Text node. Its value corresponds to the
104   name of the input Place.
105   rule Place {
106     from
107       n : PNML!NetContentElement
108       (
109          n.oclIsKindOf(PNML!Place)
110       )
111     to
112       place : XML!Element
113       (
114          name <- 'place',
115          value <- n.name.labels.first().text,
116          parent <- thisModule.resolveTemp(thisModule.getRoot(),
117   'document_net'),
118          children <- Sequence{place_id, place_name}
119       ),
120       place_id : XML!Attribute
121       (
122          name <- 'id',
123          value <- n.id,
124          parent <- n
125       ),
126       place_name : XML!Element
127       (
128          name <- 'name',
129          parent <- n,
130          children <- place_text
131       ),
132       place_text : XML!Element
133       (
134          name <- 'text',
135          parent <- place_name,
136          children <- place_xml_text
137       ),
138       place_xml_text : XML!Text
```

```
139         (
140           value <- n.name.labels.first().text,
141           parent <- place_text
142         )
143
144   }
145
146   -- The Transition rule generates three XML Elements, one XML Attribute and
147   one XML Text for each PNML Transition input element.
148   -- The first generated Element, “transition”, is a “transition” tag which
149   accepts an “id” Attribute as well as a child “name” Element. The value of
150   the “id” attribute corresponds to the one of the PNML Transition.
151   -- The generated “name” Element accepts a “text” Element as child. This
152   last one has a child which is a Text node. Its value corresponds to the
153   name of the input Transition.
154   rule Transition {
155     from
156       n : PNML!NetContentElement
157       (
158         n.oclIsKindOf(PNML!Transition)
159       )
160     to
161       transition : XML!Element
162       (
163         name <- 'transition',
164         value <- n.name.labels.first().text,
165         parent <- thisModule.resolveTemp(thisModule.getRoot(),
166   'document_net'),
167         children <- Sequence{transition_id, transition_name}
168       ),
169       transition_id : XML!Attribute
170       (
171         name <- 'id',
172         value <- n.id,
173         parent <- n
174       ),
175       transition_name : XML!Element
176       (
177         name <- 'name',
178         parent <- n,
179         children <- transition_text
180       ),
181       transition_text : XML!Element
182       (
183         name <- 'text',
184         parent <- transition_name,
185         children <- transition_xml_text
186       ),
187       transition_xml_text : XML!Text
188       (
189         value <- n.name.labels.first().text,
190         parent <- transition_text
191       )
192
```

```
193   }
194
195   -- The Arc rule generates three XML Elements, three XML Attributes and one
196   XML Text for each PNML Arc input element.
197   -- The generated Element is an “arc” tag that has three Attribute children:
198   “id”, “source” and “target”, as well as a child “name” Element. The value
199   of the “id” attribute corresponds to the one of the PNML Arc. Values of the
200   “source” and “target” attributes respectively correspond to the id of the
201   source and the id of the target of the input Arc.
202   -- The generated “name” Element accepts a “text” Element as child. This
203   last one has a child which is a Text node. Its value corresponds to the
204   name of the input Transition.
205   rule Arc {
206     from
207       n : PNML!Arc
208     to
209       arc : XML!Element
210       (
211         name <- 'arc',
212         value <- n.name.labels.first().text,
213         parent <- thisModule.resolveTemp(thisModule.getRoot(),
214   'document_net'),
215         children <- Sequence {arc_name, arc_id, source, target}
216       ),
217       arc_id : XML!Attribute
218       (
219         name <- 'id',
220         value <- n.id,
221         parent <- n
222       ),
223       arc_name : XML!Element
224       (
225         name <- 'name',
226         parent <- n,
227         children <- arc_text
228       ),
229       arc_text : XML!Element
230       (
231         name <- 'text',
232         parent <- arc_name,
233         children <- arc_xml_text
234       ),
235       arc_xml_text : XML!Text
236       (
237         value <- n.name.labels.first().text,
238         parent <- arc_text
239       ),
240       -- source and target attribute are initialised by the id of the
241   element pointed
242       source : XML!Attribute
243       (
244         name <- 'source',
245         value <- n.source.id,
246         parent <- n
```

```
247         ),
248         target : XML!Attribute
249         (
250           name <- 'target',
251           value <- n.target.id,
252           parent <- n
253         )
254     }
```

3.3.1.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 14 and Figure 15, there is one input metamodel (PNML) and one output (XML). In Path Editor, place in "PNML" the path of the PNML metamodel; do the same for "XML". In "IN" place the path of an Ecore file (a model conforming to our PNML metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the XML metamodel).



Figure 14 - PNML to XML configuration - part one



Figure 15 - PNML to XML configuration - part two

3.3.2. XML to PNML Transformation: Injector

3.3.2.1. Description of the Transformation

The ATL code for the Grafcet to Petri Net transformation consists of 5 rules and 3 helpers.

Helpers:

- The first helper **getAttrVal**, returns the value of an attribute (identified by its name, passed as a parameter) of the contextual XML Element. For this purpose, its collects, among the children of this contextual Element, the Attribute whose name matches the name passed in parameter. The helper returns the value of the first matched attribute.
- The **getName** helper returns the name of a “net” or a “place” XML Element. To this end, it first gets, among its Element children, the one named “name”. It then gets the “text” XML Element child of this new node, and finally returns the value associated with it.
- The **getLink** helper collects all instances of xml element and search the one whose id matches the id passed in parameter. The helper returns the first xml element of the collection.

Rules:

- The **PNMLDocument** rule generates a PNMLDocument from the input XML Root Element.
- The **Net** rule generates a NetElement from each “net” XML Element input element. The name of the generated NetElement is computed by calling the **getName** helper. Its set of Places, Transitions and Arcs are initialized by the other rules. The link to its parent, the PNMLDocument, is also created.
- The **Place** rule generates a PNML Place for each “place” XML Element. The name of the generated Place is computed by a call to the **getName** helper. Its id is copied from the one of the input XML Element. The link to its parent, the NetElement, is also created.
- The **Transition** rule generates a PNML Transition for each “transition” XML Element. The name of the generated Transition is computed by a call to the **getName** helper. Its id is copied from the one of the input XML Element. The link to its parent, the NetElement, is also created.
- The **Arc** rule generates a PNML Arc for each “arc” XML Element. The name of the generated Arc is computed by a call to the **getName** helper. Its id is copied from the one of the input XML Element. Its source (obtained by means of the **getLink** helper) corresponds to the XML Element which id is contained in the child attribute named “source”. Idem for the target. The link to its parent, the NetElement, is also created.

3.3.2.2. ATL Code

```
1  module XML2PNML;
2  create OUT : PNML from IN : XML;
3
4  -- The getAttrVal helper, returns the value of an attribute (identified by
5  its name, passed as a parameter) of the contextual XML Element.
6  -- For this purpose, its collects, among the children of this contextual
7  Element, the Attribute whose name matches the name passed in parameter.
8  -- The helper returns the value of the first matched attribute.
9  -- CONTEXT: XML!Element
10 -- RETURN: String
11 helper context XML!Element def: getAttrVal(name : String) : String =
12    let a : Sequence(XML!Attribute) = self.children->select(c |
13 c.oclIsTypeOf(XML!Attribute) and c.name = name) in
14    if a.isEmpty() then
15       ''
16    else
17       a.first().value
18    endif;
19
20 -- The getName() helper returns the name of a “net” or a “place” XML
21 Element.
22 -- To this end, it first gets, among its Element children, the one named
23 “name”.
24 -- It then gets the “text” XML Element child of this new node, and finally
25 returns the value associated with it.
26 -- CONTEXT: XML!Element
27 -- RETURN: String
28 helper context XML!Element def : getName() : String =
```

```
29      self.children->select(c | c.oclIsTypeOf(XML!Element) and c.name =
30    'name')->first().children
31          ->select(d | d.oclIsTypeOf(XML!Element) and d.name = 'text')-
32    >first().children
33          ->select(e | e.oclIsKindOf(XML!Text))->first().value;
34
35    -- The getLink helper, collects all instances of xml element and search the
36    one whose id matches the id passed in parameter.
37    -- The helper returns the first xml element of the collection.
38    -- CONTEXT: n/a
39    -- RETURN: XML!Element
40    helper def: getLink(id : String) : XML!Element =
41      XML!Element.allInstances()->select(z | z.getAttrVal('id') = id)->first();
42
43
44    -- The PNMLDocument rule generates a PNMLDocument from the input XML Root
45    Element.
46    rule PNMLDocument {
47      from
48        x : XML!Root
49      to
50        document : PNML!PNMLDocument
51        (
52          xmlns <- uri
53        ),
54        uri : PNML!URI
55        (
56          value <- x.getAttrVal('xmlns')
57        )
58    }
59
60    -- The Net rule generates a NetElement from each “net” XML Element input
61    element.
62    -- Name of the generated NetElement is computed by calling the getName
63    helper.
64    -- Its set of Places, Transitions and Arcs are initialized by the other
65    rules.
66    -- The link to its parent, the PNMLDocument, is also created.
67    rule Net {
68      from
69        x : XML!Element
70        (
71          x.name = 'net'
72        )
73      to
74        net_element : PNML!NetElement
75        (
76          name <- named_element,
77          type <- type_uri,
78          -- pointer on the root element
79          document <- x.parent
80        ),
81        type_uri : PNML!URI
82        (
```

```
83          value <- x.getAttrVal('type')
84       ),
85       named_element : PNML!Name
86       (
87          labels <- label
88       ),
89       label : PNML!Label
90       (
91          text <- x.getName()
92       )
93    }
94
95    -- The Place rule generates a PNML Place for each “place” XML Element.
96    -- Name of the generated Place is computed by a call to the getName helper.
97    -- Its id is copied from the one of the input XML Element.
98    -- The link to its parent, the NetElement, is also created.
99    rule Place {
100      from
101         x : XML!Element
102         (
103            x.name = 'place'
104         )
105      to
106         n : PNML!Place
107         (
108            name <- named_element,
109            -- pointer on the net element
110            net <- x.parent,
111            id <- x.getAttrVal('id'),
112            location <- ''
113         ),
114         named_element : PNML!Name
115         (
116            labels <- label
117         ),
118         label : PNML!Label
119         (
120            text <- x.getName()
121         )
122   }
123
124   -- The Transition rule generates a PNML Transition for each “transition”
125   XML Element.
126   -- Name of the generated Transition is computed by a call to the getName
127   helper.
128   -- Its id is copied from the one of the input XML Element.
129   -- The link to its parent, the NetElement, is also created.
130   rule Transition {
131      from
132         x : XML!Element
133         (
134            x.name = 'transition'
135         )
136      to
```

```
137         n : PNML!Transition
138         (
139           name <- named_element,
140           -- pointer on the net element
141           net <- x.parent,
142           id <- x.getAttrVal('id')
143         ),
144         named_element : PNML!Name
145         (
146           labels <- label
147         ),
148         label : PNML!Label
149         (
150           text <- x.getName()
151         )
152   }
153
154   -- The Arc rule generates a PNML Arc for each “arc” XML Element.
155   -- Name of the generated Arc is computed by a call to the getName helper.
156   -- Its id is copied from the one of the input XML Element.
157   -- Its source (obtained by means of the getLink helper) corresponds to the
158   XML Element which id is contained in the child attribute named “source”.
159   Idem for the target.
160   -- The link to its parent, the NetElement, is also created.
161   rule Arc {
162     from
163         x : XML!Element
164         (
165           x.name = 'arc'
166         )
167     to
168         n : PNML!Arc
169         (
170           name <- named_element,
171           id <- x.getAttrVal('id'),
172           net <- x.parent,
173           -- seek of the element pointed by the source id contained in the xml
174   file
175           source <- thisModule.getLink(
176             (x.children->select(c | c.oclIsKindOf(XML!Attribute) and c.name =
177   'source')->first().value)
178           ),
179           -- seek of the element pointed by the target id contained in the xml
180   file
181           target <- thisModule.getLink(
182             (x.children->select(c | c.oclIsKindOf(XML!Attribute) and c.name =
183   'target')->first().value)
184           )
185
186         ),
187         named_element : PNML!Name
188         (
189           labels <- label
190         ),
```

```
191         label : PNML!Label
192         (
193            text <- x.getName()
194         )
195     }
196
```

3.3.2.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 16 and Figure 17, there is one input metamodel (XML) and one output (PNML). In Path Editor, place in "XML" the path of the XML metamodel; do the same for "PNML". In "IN" place the path of an Ecore file (a model conforming to our XML metamodel in Ecore format), and in "OUT" the path for the results (the generated file is an Ecore file conforming to the PNML metamodel).



Figure 16 - XML to PNML configuration - part one



Figure 17 - XML to PNML configuration - part two

3.3.3. XML to PNML text (Extract XML)

3.3.3.1. Description of the Transformation

The ATL code, that allows generating a PNML valid and well-formed XML text file from an XML model, for this transformation consists in 4 helpers and 1 query.

The aim of this query is to extract each of the elements that compose the input XML model into an output XML file. Contrary to rules that are implemented to generate a model from another model, a query allows calculating output text files from an input model (see [3]). This is the reason why we need to use queries for this type of transformation: generating an XML file from an XML model.

The implemented query get the Root element of the XML model and call the “toString2()” helper on it. The content is generated by the “toString2()” helper called on the Root element of the XML model.

There are three “toString2()” helpers with different contexts. The XML!Attribute one simply returns the name and the value of an attribute in the correct string format. The XML!Text one only returns the string value contained in a text node. The XML!Element one returns the valid and well-formed content of the output XML file by parsing recursively all the element of the input XML model. Note that it sometimes calls the XML!Attribute and XML!Text “toString2()” helpers.

3.3.3.2. ATL Code

```
1  query XML2Text = XML!Root.allInstances()
2     ->asSequence()
```

```
3     ->first().toString2('').writeTo('C:\\... Complete this path ...\\Grafcet-
4   PetriNet\\Models\\XML2Text_example.xml');
5
6   helper context XML!Element def: toString2(indent : String) : String =
7     let na : Sequence(XML!Node) =
8       self.children->select(e | not e.oclIsKindOf(XML!Attribute)) in
9     let a : Sequence(XML!Node) =
10      self.children->select(e | e.oclIsKindOf(XML!Attribute)) in
11    indent + '<' + self.name +
12    a->iterate(e; acc : String = '' |
13      acc + ' ' + e.toString2()
14    ) +
15    if na->size() > 0 then
16      '>'
17      + na->iterate(e; acc : String = '' |
18        acc +
19        if e.oclIsKindOf(XML!Text) then
20          ''
21        else
22          '\r\n'
23        endif
24        + e.toString2(indent + '  ')
25      ) +
26      if na->first().oclIsKindOf(XML!Text) then
27        '</' + self.name + '>'
28        else
29          '\r\n' + indent + '</' + self.name + '>'
30      endif
31    else
32      '/>'
33    endif;
34
35  helper context XML!Attribute def: toString2() : String =
36    self.name + '=\"' + self.value + '\"';
37
38  helper context XML!Text def: toString2() : String =
39    self.value;
```

3.3.3.3. Configuration of the Transformation

As illustrated by the transformation configuration's Figure 18 and Figure 19, there is one input metamodel (XML). In Path Editor, place in "XML" the path of the XML metamodel. In "IN"place the path of an Ecore file (a model conforming to our XML metamodel in Ecore format).

The generated file is an Ecore file conforming to the XML metamodel. This file does not appear in the configuration, it is defined in the ATL code of the transformation. So in the XML to Text ATL file, ensure that the output file path is correct at the top of the file (Figure 20).



Figure 18 - XML to Text configuration - part one



Figure 19 - XML to Text configuration - part two



Figure 20 - XML to Text ATL file

I. Grafcet metamodel in KM3 format

```
package Grafcet {
   abstract class LocatedElement {
      attribute location : String;
   }
   abstract class NamedElement extends LocatedElement {
      attribute name : String;
   }
   class Grafcet extends NamedElement {
      reference elements[*] container : Element oppositeOf grafcet;
      reference connections[*] container : Connection oppositeOf grafcet;
   }
   -- @begin elements
   abstract class Element extends NamedElement {
      reference grafcet : Grafcet oppositeOf elements;
   }
   class Step extends Element {
      attribute isInitial : Boolean;
      attribute isActive : Boolean;
      attribute action : String;
      reference incomingConnections[*] : TransitionToStep oppositeOf to;
      reference outgoingConnections[*] : StepToTransition oppositeOf from;
   }
   class Transition extends Element {
      attribute condition : String;
      reference incomingConnections[*] : StepToTransition oppositeOf to;
      reference outgoingConnections[*] : TransitionToStep oppositeOf from;

   }
   -- @end elements
   --@begin connections
   abstract class Connection extends NamedElement {
      reference grafcet : Grafcet oppositeOf connections;
   }
   class StepToTransition extends Connection {
      reference from : Step oppositeOf outgoingConnections;
      reference to : Transition oppositeOf incomingConnections;
   }
   class TransitionToStep extends Connection {
      reference from : Transition oppositeOf outgoingConnections;
      reference to : Step oppositeOf incomingConnections;
   }

   --@end connections
}
package PrimitiveTypes {
   datatype String;
   datatype Boolean;
}
```

II. Petri Net metamodel in KM3 format

```
package PetriNet {
   abstract class LocatedElement {
      attribute location : String;
   }
   abstract class NamedElement extends LocatedElement {
      attribute name : String;
   }
   -- @comment top element
   class PetriNet extends NamedElement {
      reference elements[*] container : Element oppositeOf net;
      reference arcs[*] container : Arc oppositeOf net;
   }
   -- @begin elements
   abstract class Element extends NamedElement {
      reference net : PetriNet oppositeOf elements;
   }
   class Place extends Element {
      reference incomingArc[*] : TransitionToPlace oppositeOf to;
      reference outgoingArc[*] : PlaceToTransition oppositeOf from;
   }
   class Transition extends Element {
      reference incomingArc[1-*] : PlaceToTransition oppositeOf to;
      reference outgoingArc[1-*] :  TransitionToPlace oppositeOf from;
   }
   -- @end elements
   --@begin arcs
   abstract class Arc extends NamedElement {
      attribute weight : Integer;
      reference net : PetriNet oppositeOf arcs;
   }
   class PlaceToTransition extends Arc {
      reference from : Place oppositeOf outgoingArc;
      reference to : Transition oppositeOf incomingArc;
   }
   class TransitionToPlace extends Arc {
      reference from : Transition oppositeOf outgoingArc;
      reference to : Place oppositeOf incomingArc;
   }
   --@end arcs
}
package PrimitiveTypes {
   datatype String;
   datatype Integer;
}
```

III. PNML metamodel in KM3 format

```
package PNML {
   abstract class LocatedElement {
      attribute location : String;
   }
   abstract class IdedElement extends LocatedElement {
      attribute id : String;
   }
   -- @begin declaration of types
   class URI extends LocatedElement {
      attribute value : String;
   }
   -- @end declaration of types
   -- @comment single top element (like in XML document)
   class PNMLDocument extends LocatedElement {
      reference xmlns container : URI;
      reference nets[1-*] container : NetElement oppositeOf document;
   }
   -- @comment a petri net element
   class NetElement extends IdedElement {
      -- @comment typer reference the PNTD associed with the net
      reference type container : URI;
      reference document : PNMLDocument oppositeOf nets;
      reference contents[*] container : NetContent oppositeOf net;
      reference name[0-1] container : Name oppositeOf net;
   }
   -- @comment content of a petri net element
   abstract class NetContent extends LocatedElement {
      reference net : NetElement oppositeOf contents;
      reference name[0-1] container : Name oppositeOf netContent;
   }
   -- @comment element used for abstraction (Name, Inscription and InitialMarking)
   abstract class LabeledElement extends LocatedElement {
      reference labels[*] container : Label oppositeOf labeledElement;
   }
   class Label extends LocatedElement {
      attribute text : String;
      reference labeledElement : LabeledElement oppositeOf labels;
   }
   class Name extends LabeledElement {
      reference net[0-1] : NetElement oppositeOf name;
      reference netContent[0-1] : NetContent oppositeOf name;
   }
   -- @comment element used for abstraction (Place and Transition)
   abstract class NetContentElement extends NetContent,IdedElement {
   }
   class Arc extends NetContent,IdedElement {
      reference source : NetContentElement;
      reference target : NetContentElement;
   }
   class Place extends NetContentElement {
   }
   -- @comment a transition element
   class Transition extends NetContentElement {
   }
}
package PrimitiveTypes { datatype String; }
```

IV. XML metamodel in KM3 format

```
package XML {
   abstract class Node {
      attribute startLine[0-1] : Integer;
      attribute startColumn[0-1] : Integer;
      attribute endLine[0-1] : Integer;
      attribute endColumn[0-1] : Integer;
      attribute name : String;
      attribute value : String;
      reference parent[0-1] : Element oppositeOf children;
   }
   class Attribute extends Node {}
   class Text extends Node {}
   class Element extends Node {
      reference children[*] ordered container : Node oppositeOf parent;
   }
   class Root extends Element {}
}
package PrimitiveTypes {
   datatype Boolean;
   datatype Integer;
   datatype String;
}
```

V. References

[1] The Petri Net Markup Language (PNML). Documentation and tools available at http://www.informatik.huberlin.de/top/pnml/about.html.

[2] KM3: Kernel MetaMetaModel. Available at http://dev.eclipse.org/viewcvs/indextech.cgi/~checkout~/gmthome/doc/atl/index.html.

[3] ATL User manual, “4.1 Queries and the Generation of Text” subsection, http://www.eclipse.org/gmt/, ATL subproject, ATL Documentation Section